
1 INTRODUCTION

Paradynis a parallel performance measurement tool that performsdynamic instrumentationon
running applications. Instrumentation is a sequence of instructions, such as a code fra
inserted at the beginning of a function to increment a count of the number of times the funct
invoked. The instructions are compiled from a source language, called theMetric Description
Language(MDL). For Paradyn to properly instrument (and thus measure) your application’s
formance, you must specify all the metrics you will need using MDL in aParadyn configuration
file. Commonly used metrics have already been programmed for you as part of the Paradyn
bution in a file namedparadyn.rc .

This paper is written to help you become more proficient in programming with MDL. Spe
ically, this paper is designed to save you time and frustrations in using this new language. M
unlike any language that you might have used, so we want to ease your introduction. We a
you have already familiarized yourself with the MDL language syntax: if not, please refer to
Paradyn User’s Guide. We further assume you are proficient in programming; MDL is
unusual programming language, as you will shortly discover, and understanding advance
gramming concepts will ease your learning.

This paper walks you through from the bottom up. We first start with the basic MDL me
data types: counters and timers. The rest of the paper essentially goes through tips on
properly use them. Even seasoned programmers will be surprised by how subtle the pro
ming issues are when using even seemingly simple concepts such as counters and timers
the counters and timers do not update themselves automatically, we go on to actual instru
tion code that updates them.

Three types of instrumentation can be specified in MDL, with each type named by their
tion (or point of insertion): entry-point, return-point, and callsite. Entry-point instrumentation
inserted at the beginning of functions. Return-point instrumentation is inserted at the end of
tions. Callsite instrumentation is inserted at places where functions call other functions
dealing with exclusive metrics. Entry-point and return-point instrumentation are closely rel
so we consider them together in one section. Callsite instrumentation is described in a se
section.

Last, we devote a section on miscellaneous issues that did not fit nicely into any of the a
mentioned sections and did not warrant separate sections themselves.

To simplify the presentation, we will be providing only MDL code fragments, and not co
plete metric implementations. In addition, thoughconstraintsare key to instrumentation, none o
the MDL examples areconstrained . That is, all examples will be written as if for whole-pro
gram metrics. Again, this is done to simplify the text. The metrics you write, however, will u
ally need to beconstrained .

2 COUNTERS

Counters are one of two metric data types supported by MDL. The second, timers, will be
ered in Section 3.
MDL Programmer’s Guide May 28, 2003 Release 4.0

2.1 Counters are initialized to zero

Therefore, if you are using a counter as a flag, try NOT to use zero as the “okay” value.

Here is a fragment of a metric that counts the number of I/O function calls,exceptthose calls
made from within MPI functions. (MPI is a library of functions used for exchanging values in
allel programs.)

  // Good example of using a counter as a flag

  counter not_in_mpi;  // Flag is 1 when program is outside  an MPI function

  foreach func in mpi_funcs {
    append  preInsn func.entry  (* not_in_mpi = 0; *)
    prepend preInsn func.return (* not_in_mpi = 1; *)
  }

  foreach func in io_funcs {
    append preInsn func.entry (* if (not_in_mpi == 1) io_ops++; *)
  }

The not_in_mpi flag is set only after a return from an MPI function, and we increment
io_ops counter only if the flag is on. That is, we are incrementing the counter of I/O func
calls only when we are sure we are not executing within an MPI function. Assuming that the
functions do not call each other, directly or indirectly, it is impossible for thenot_in_mpi flag to
have a value of 1 when within an MPI function. To show this, we need a case analysis. A
moment of instrumentation, the program is either executing an MPI function or it is not, so
are only two cases to consider. (Note that if a program is currently within a function a
moment of instrumentation, any new entry-point instrumentation of the function may not be
cuted for the current invocation of the function.)

Case 1:  Program is inside an MPI function at the moment of instrumentation.
Sincenot_in_mpi  is initialized to zero, the flag has the correct off value.  As a result, we
will not mistakenly increment the I/O function calls count when inside an MPI function.

Case 2:  Program is outside an MPI function at the moment of instrumentation.
Sincenot_in_mpi is initialized to zero, the flag has the incorrect off value. However, this m
ric was written for an MPI application, so this situation will be corrected shortly at the ne
call of an MPI function.
Suppose we mistakenly used zero as the “okay” flag value for this metric, as in the follo

example.

  // Bad example of using a counter as a flag

  counter in_mpi;  // Flag is 1 when program is inside  an MPI function

  foreach func in mpi_funcs {
    append  preInsn func.entry  (* in_mpi = 1; *)
    prepend preInsn func.return (* in_mpi = 0; *)
MDL Programmer’s Guide May 28, 2003 Release 4.0

  foreach func in io_funcs {
    append preInsn func.entry (* if (in_mpi == 0) io_ops++; *)
  }

Using a case analysis, we find that it is possible to mistakenly count I/O function calls m
from within MPI functions.

Case 1:  Program is inside an MPI function at the moment of instrumentation.
Sincein_mpi  is initialized to zero, the flag has the incorrect off value.  As a result, we
might mistakenly increment the I/O function calls count when inside the currently execu
MPI function.

Case 2:  Program is outside an MPI function at the moment of instrumentation.
Sincein_mpi  is initialized to zero, the flag has the correct off value.

You may notice that deciding on the “okay” value of a flag depends on the errors the m
must avoid. In the above example, it was required that I/O function calls by MPI functions n
counted, so we had to choose 1 as the “okay” value.  However, there may be situations wh
must use 0 as the “okay” value.

3 TIMERS

Timers are more complex than counters, but they still can be used with a few simple rules.
dyn has four timer functions:startProcessTimer() andstopProcessTimer() for virtual clock
timers, andstartWallTimer() and stopWallTimer() for wall clock timers. For clarity, we
shall sometimes use the abbreviationsstartTimer()  andstopTimer() .

3.1 EverystartTimer()  must be matched by astopTimer()

This is similar to the syntax of many programming languages. In Pascal, aBEGIN is ended by the
first unmatchedEND. In C, an open brace{ is ended by the first unmatched close brace} . In Lisp,
an open parenthesis( is ended by the first unmatched close parenthesis) . In MDL, a start-

Timer(T) , for some timerT, is ended by the first unmatchedstopTimer(T) . An unmatched
stopTimer() has no effect. CallingstartTimer() on an already running timer neither restar
the timer nor starts a new copy of the timer. However, the timer is actually stopped ONLY w
everystartTimer()  has been matched by astopTimer() .

Unlike the Pascal, C, and Lisp examples, MDL timer matching is done as the instrumen
is executed, and not when the metric was written by you. That is, unmatched elements in P
C, and Lisp will be caught by a compiler, but unmatched MDL timers will result in incorrect tim
values. At this point, we just want you to be aware of the above rule. The later sections on i
mentation will show you how to keepstartTimer()  andstopTimer()  matched.
MDL Programmer’s Guide May 28, 2003 Release 4.0

3.2 Useappend /append  instead ofappend /prepend  for timers

Software instrumentation inherently perturbs the original program, so the program’s ti
behavior changes. Furthermore, subsequent instrumentation can change the timing beha
previous instrumentation. The following example shows how relative errors in timers can be
imized by careful relative placement of timer instrumentation blocks. Let us illustrate thi
examining instrumentation that times the execution of an I/O function.

We start with an I/O function that has already been instrumented as follows.E1 andE2 are
two pieces of entry-point instrumentation.R1 andR2 are two pieces of return-point instrumenta
tion.

Now we want to insert our timer instrumentation, written in MDL as follows.

  // The most obvious way to implement an I/O function timer metric

  foreach func in io_funcs {
    append  preInsn func.entry  (* startWallTimer(T); *)
    prepend preInsn func.return (* stopWallTimer(T);  *)
  }

The following figure shows how things would look after we have actually instrumented
I/O function with the code given above.

E1 E2

R1 R2

I/O Func

E1 E2

R1 R2

I/O Func startWallTimer(T)

stopWallTimer(T)
MDL Programmer’s Guide May 28, 2003 Release 4.0

Notice that for a single timer metric, this is the best placement of the instrumentation.
append ing startWallTimer(T) , we avoid including the execution times of the existing entr
point instrumentation. Byprepend ing stopWallTimer(T) , we avoid including the execution
times of the existing return-point instrumentation. However, let us now consider what hap
when multiple timers are inserted. For example, we may have inserted one timer to keep t
time for the entire program, one timer to keep the I/O time for amodule(a set of procedures), and
one timer to keep the I/O time for a particular procedure.

Below we have instrumentation for timersT1, T2, andT3; inserted, in that order. For the sak
of clarity, let us assume there was no preexisting instrumentation on the I/O function.

Notice thatT3 gets the best placement for being inserted last. Meanwhile,T1 gets the worst
placement after being inserted first. As new timers are inserted, the earlier timers get less a
accurate because they must also timeall the new timers’ instrumentation.

To reduce the rate of decrease in timer accuracies, we can useappend /append instead of
append /prepend  in our metrics.  That is, the MDL description becomes the following.

  // An I/O function timer metric that causes less interference

  foreach func in io_funcs {
    append preInsn func.entry  (* startWallTimer(T); *)

append  preInsn func.return (* stopWallTimer(T);  *)
  }

Here again we have instrumentation for timersT1, T2, andT3; inserted, in that order. How-
ever, this time, the return-point instrumentation has beenappend ed, instead ofprepend ed. As
new timers are inserted, the earlier timers still get less accurate, but the rate of decrease in
racy has been lessened because they must time onlyhalf the new timers’ instrumentation. Also
notice that if all the instrumentation pieces take the same amount of time to execute, each
will see the same amount of error. This may be better because then you won’t have metric
agreeing with each other about the amount of time this I/O function takes.

I/O Func startTimer( T1) startTimer( T2) startTimer( T3)

stopTimer( T3) stopTimer( T2) stopTimer( T1)
MDL Programmer’s Guide May 28, 2003 Release 4.0

Using prepend /prepend also has the same relative-error minimization property, but redu
the chances of instrumentation being executed at the earliest possible opportunity. To see t
following figure illustrates the situation after instrumentation forT1 and T2 have been inserted
with prepend /prepend . Suppose the program is executing thestartTimer(T2) instrumentation
when we need to insert instrumentation forT3. SincestartTimer(T3) will be prepend ed in
front of startTimer(T2) , startTimer(T3) will not get executed for this invocation of the I/O
function. However, had we instead beenappend ing, thestartTimer(T3) will get executed for
this invocation.

4 ENTRY-RETURN INSTRUMENTATION

In the previous sections, we looked at the basic properties of counters and timers. In this s
we look at how we can properly update the counters and timers in our entry-point and return
instrumentation.  Section 5 will cover callsite instrumentation.

4.1 Return-point execution doesNOT imply entry-point execution

The simplest example is the case where the program was executing within the function wh
function becomes instrumented. The entry-point instrumentation will not get executed fo
invocation of the function because the entry point has already been passed. However, the

I/O Func startTimer( T1) startTimer( T2) startTimer( T3)

stopTimer( T1) stopTimer( T2) stopTimer( T3)

I/O Func startTimer( T2) startTimer( T1)

stopTimer( T2) stopTimer( T1)
MDL Programmer’s Guide May 28, 2003 Release 4.0

illustrate how to handle this situation.

Here are the most obvious MDL statements for an inclusive constraint flag.constraintFlag

is nonzero if there is any known active invocation of the instrumented function.

  counter constraintFlag;

  append preInsn func.entry  (* constraintFlag += 1; *)
  append preInsn func.return (* constraintFlag -= 1; *)

Let us make a simple walk-through of the above MDL statements. As with all counters,con-

straintFlag is initialized to zero. On entry to the instrumented function,constraintFlag is
incremented to 1. On return from the instrumented function,constraintFlag is decremented
back to 0. Should the instrumented function be recursive, the recursive invocation will
increaseconstraintFlag on entry, but will properly restoreconstraintFlag on return. It
seems that the above MDL statements correctly giveconstraintFlag a nonzero value when
there is an active invocation of the function.

However, suppose that the program was executing within the function at the moment of in
mentation. Then the entry-point instrumentation will not be executed, but the return-point in
mentation is. The result is thatconstraintFlag gets set to -1 at return. NowconstraintFlag is
nonzero even when there is no active invocation of the function. It gets worse. On the next
cation of the function, the entry-point instrumentation will increaseconstraintFlag to zero, and
the return-point instrumentation will decreaseconstraintFlag back to -1. Therefore,con-

straintFlag is zero when there is a known active invocation of the function. If this function
nonrecursive, our instrumentation has failed in every case.

To correct the situation, here is a better MDL implementation.constraintFlag is nonzero if
and only if it has been incremented at entry to the function. Using this knowledge, the new im
mentation executes the return-point instrumentation only if the entry-point instrumentation
been executed. Another way to look at it is that we simply do not decrementconstraintFlag

below zero. Of course, if instrumentation occurred while the program was executing withi
function,constraintFlag may still have the incorrect value of zero for the duration of the fun
tion’s current invocation.

  // Being careful in return-point instrumentation

  counter constraintFlag;

  append preInsn func.entry  (* constraintFlag += 1; *)
append preInsn func.return (* if (constraintFlag != 0) constraintFlag -= 1;*)

4.2 Entry-point execution doesNOT imply return-point execution

Some functions will return abnormally. For example, suppose there was an exception. Som
guages with exception handling will allow a function to unwind the callstack to an arbitrary d
until it finds a function that can handle the exception. Another example is thelongjmp() library
MDL Programmer’s Guide May 28, 2003 Release 4.0

function from the C programming language. Calling this function essentially unwinds the
stack. Yet another example is a case where we did not detect an instruction sequence as
tion-return sequence. There are many ways to return from a function. With optimizing comp
and creative assembly programmers, we do not claim to know all the instruction sequence
may be used to return from a function. As a result, we may not have instrumented all r
points of a function.

Let us look at a timer example. Here are the most obvious MDL statements to measu
execution time of a function. TimerT is on if and only if there is a known active invocation of th
function.

  append preInsn func.entry  (* startProcessTimer(T); *)
  append preInsn func.return (* stopProcessTimer(T);  *)

Let us take a simple walk-through of the above code. On entry to the function, timerT is
started. On return from the function, timerT is stopped. If the program was executing within th
function at the moment of instrumentation, then we will simply stop a timer that was not sta
which has no effect.  Things look good so far.

However, suppose there was a normal entry to the function (timerT started) followed by a
rarely used exceptional return (timerT not stopped). Remember that timers are stopped only if
startTimer() calls are matched bystopTimer() calls. In this case, thestartProcess-

Timer(T) at entry was not matched by thestopProcessTimer(T) at return, so timerT is still
running. On the next invocation of the function, we execute a newstartProcessTimer(T) , so
we now have two unmatchedstartProcessTimer(T) . At normal return, we executestopPro-

cessTimer(T) to match the newstartProcessTimer(T) , but the other unmatchedstartPro-

cessTimer(T) is still unmatched. So at return, the timerT is still running. In fact, timerT will
always remain running, regardless of whether or not the function is actually being executed

To solve this problem, we write a better MDL implementation that has a “self-healing” p
erty. The next normal execution of the function will correct the instrumentation problems o

previous exceptional execution of the function1. We accomplish this by making sure that there
never more than one unmatchedstartProcessTimer(T) . The new MDL code follows. (Unfor-
tunately, there is no simple way to extend this technique to recursive functions.)

  // A self-healing timer metric

  counter T_is_running ;  // Flag is 1 if timer T is running

  append preInsn func.entry
    (* if ( T_is_running  == 0) startProcessTimer(T);

T_is_running  = 1;
     *)

  append preInsn func.return
    (* T_is_running  = 0;
       stopProcessTimer(T);

1. Note, however, that this “healing” of the instrumentation problem such that subsequent execution will b
correct, doesn’t correct or compensate for spurious accounting which takes place in the interrim.
MDL Programmer’s Guide May 28, 2003 Release 4.0

We are using the counterT_is_running as a flag that is true when there is an unmatch
startProcessTimer(T) . T_is_running is correctly initialized to zero since the timerT is not
running at the moment of instrumentation. At entry to the function, we turn the timer on if
only if it has not already been turned on. Then we setT_is_running to 1 to indicate that at this
point we are sure the timer is running. Notice that no matter how many times you repeatedl
cute this entry-point instrumentation, there is at most onestartProcessTimer(T) that actually
gets executed. At normal return from the function, thestopProcessTimer(T) will match the sin-
gle unmatchedstartProcessTimer(T) . We also setT_is_running to zero since we are sure
there will be no unmatchedstartProcessTimer(T) .

Let us look at the case where we enter the function normally, executestartProcess-

Timer(T) , and return exceptionally without executing the return-point instrumentation. Now
timer T is incorrectly running, and theT_is_running flag is still set to 1. On the next invocation
of the function, thestartProcessTimer(T) is not executed. If the function returns normally th
time, thestopProcessTimer(T) will correctly turn off the timer, and theT_is_running flag will
be correctly set to 0.  Our instrumentation is now “healed”.

4.3 Avoid using entry-point post-instrumentation

The sequence in which Paradyn executes instructions at an instrumentation point is as fo
First, any pre-instrumentation is executed, then the program instructions located at the instr
tation point, and afterwards, any post-instrumentation is executed.  This is illustrated below

You should try to avoid using entry-point post-instrumentation because the program ins
tions found at the point may be branch instructions. These instructions will divert execution
away from the post-instrumentation, as illustrated in the figure below. When this happens
post-instrumentation will not be executed. In practice, we have not found it necessary t
entry-point post-instrumentation.

program instructions at point

post1 post2

pre1 pre2
MDL Programmer’s Guide May 28, 2003 Release 4.0

Therefore, you should avoid using any MDL statements that look like the following. Fu
versions of Paradyn may even treat them as syntax errors.

  append postInsn  func. entry   ...   // RISKY!
  prepend postInsn  func. entry   ...   // Just as RISKY!

4.4 DoNOT use return-point post-instrumentation

Return-point post-instrumentation does not work and does not make sense. Therefore, P
treats any attempt to use such instrumentation as a syntax error.

  append postInsn  func. return   ...   // Syntax ERROR!
  prepend postInsn  func. return   ...   // Syntax ERROR!

branch instructions

post1 post2

pre1 pre2

diverted execution flow

postInsn ignored
MDL Programmer’s Guide May 28, 2003 Release 4.0

5 CALLSITE INSTRUMENTATION

Callsite instrumentation is commonly used in exclusive timer metrics. These metrics measu
execution time of a function, but they must exclude the time spent in functionscalledby the func-
tion being timed.  Here is a typical example of MDL code for an exclusive timer.

  counter T_is_running;  // Flag is 1 if timer T is running

  append preInsn func.entry
    (* if (T_is_running == 0) startProcessTimer(T);
       T_is_running = 1;
     *)

  append preInsn func.return
    (* T_is_running = 0;
       stopProcessTimer(T);
     *)

  foreach callsite in func.calls {
    append preInsn  callsite
      (* T_is_running = 0;
         stopProcessTimer(T);
       *)

    append postInsn callsite
      (* if (T_is_running == 0) startProcessTimer(T);
         T_is_running = 1;
       *)
  }

Let us take a simple walk-through. At entry to the function, the timer is started. At re
from the function, the timer is stopped. This times the execution of the function. Now we
ceed to exclude the execution times of functions called by this function. We accomplish th
instrumenting the callsites. At each callsite, we stop the timer before entering the called fun
and we restart the timer after returning from the called function.

Notice in the MDL code given above that the callsite pre-instrumentation is the same a
function return-point instrumentation and that the callsite post-instrumentation is the same
function entry-point instrumentation.  In general, this is true for all exclusive metrics.

5.1 In exclusive metrics,preInsn callsite  is analogous tofunc.return

At a callsite, the function being measured is going to temporarily stop executing until the c
function returns. At function return, the function is going to temporarily stop executing until
next invocation of the function. To an exclusive metric, the difference in why the function is s
ping execution is irrelevant. All that matters is that the exclusive metric must halt until the f
tion resumes execution.
MDL Programmer’s Guide May 28, 2003 Release 4.0

5.2 In exclusive metrics,postInsn callsite  is analogous tofunc.entry

Immediately following a callsite, the function being measured is going to resume execution
temporarily stopping to let another function execute. At function entry, the function is goin
resume execution after being temporarily stopped between invocations. To an exclusive m
the difference in why the function is resuming execution is irrelevant. All that matters is tha
exclusive metric must be restarted until the function stops execution again.

6 MISCELLANEOUS

This section contains MDL programming tips that could not nicely fit within any of the previ
sections and that could not warrant a separate section by themselves.

6.1 All EventCounter-style metric values must increase monotonically

The values of your EventCounter metrics should not be allowed to decrease in value, no

how slightly or for how short an amount of time.2 Paradyn rigorously checks this condition an
will immediately abort if the check ever fails. A typical example of such a metric is one that a
mulates a count of the number of bytes transferred by the I/O functions. Sample MDL co
given below.  The I/O functions give the number of bytes transferred as the return value.

  // Blindly adding function return values

  counter in_sampling;     // Flag is 0 if DYNINSTalarmExpire() is not
                           //  currently being executed

  foreach func in io_funcs {
    prepend preInsn func.return constrained
      (* if (in_sampling == 0) io_bytes += $return ; *)
  }

  foreach func in DYNINSTalarmExpire {
    prepend preInsn func.entry
      (* in_sampling = 1; *)

    append preInsn func.return
      (* in_sampling = 0; *)
  }

However, the I/O functions also return -1 if they encounter errors. In fact, it is a general p
tice in C programming for functions to return negative values to indicate errors. Therefore, w
ever an I/O function returns due to error, our instrumentation will add anegativenumber to the

2. SampledFunction-style metrics, of which an example is provided in Section 6.2, allow reported values 
change arbitrarily, however, they must still remain non-negative.
MDL Programmer’s Guide May 28, 2003 Release 4.0

metric value counterio_bytes . Paradyn will detect this decrease in the metric’s value, a
immediately abort, ending your application’s performance measurement session with it.

To solve this problem we simply check the numbers before adding them to their respe
metric value counters.  The MDL code from above is corrected below by adding such a che

  // Careful when adding function return values

  counter in_sampling;     // Flag is 0 if DYNINSTalarmExpire() is not
                           //  currently being executed

  foreach func in io_funcs {
    prepend preInsn func.return constrained
      (* if ((in_sampling == 0) && ( $return > 0 )) io_bytes += $return; *)
  }

  foreach func in DYNINSTalarmExpire {
    prepend preInsn func.entry
      (* in_sampling = 1; *)

    append preInsn func.return
      (* in_sampling = 0; *)
  }

6.2 Use SampledFunction-style metrics to track values in a user application

One use of MDL likely to be of particular interest to application or library developers, is the a
ity to query a program value and track its evolution. This is generally achieved in two steps:
ification of a query function in the user program which returns the value of interest (simila
those common for returning the values of private members of C++ classes), and a specifica
when this value should be queried when the metric is enabled.

Because Paradyn uses a shared-memory sampling approach for efficiently extracting p
mance data from the application, a suitable user function needs to be identified as the poin
values should be sampled: this may well be the function (or list of functions) which actu
update the value of interest.

The following example functions from a user program query and update a program
(which in this case is a global variable, but could have been accessed by any other means

unsigned int value;
unsigned int program_value() { return (value); }
void update_program_value (unsigned int new_value) { value=new_value; }

The metric itself should be specified of styleSampledFunction (and unitsTypesampled ),
such that the current value is available, rather than the delta from the previously sampled
This also has the advantage of allowing values to both increase and decrease, rather than
ing the value to increase monotonically as is the case with the EventCounter style of metric.
however, that sampled values need to be unsigned integers.
MDL Programmer’s Guide May 28, 2003 Release 4.0

// sample/report a program value after it is updated in a user program

resourceList update_function_list is procedure {
        items { “update_program_value” };        // user’s update function
        flavor { unix };
        library false;
}

metric programValue {
name “program_value”;
style SampledFunction;
units value_units;
unitsType sampled;
aggregateOperator sum;
flavor { unix };

  base is counter {
    foreach func in update_function_list {
      append preInsn func.return constrained
        (* programValue = program_value(); *)     // user’s sampling function
    }
  }
}

Instead of invoking the user functionprogam_value() to obtain the value of interest, MDL
provides the specially-definedreadSymbol  query function to directly read a global variable:

(* programValue = readSymbol(“value”); *) // read global symbol value

Finally, it is worth noting that while the requested value is updated and stored as specifi
the application program’s space, sampling of this value and its reporting by the Paradyn da
to the Paradyn front-end happens completely asynchronously. A consequence of this will b
Paradyn will generally miss data value updates (or repeatedly re-sample the same value) l
to spurious accumulations (in totals or averages) when compared to sample accumulations
user program itself. At best, the reported samples are an approximation of the actual s
updates requested.

6.3 Instrumenting recursive functions

Recursive functions present particular difficulties for instrumentation metrics, and a nu
of the techniques presented in this guide do not readily apply to this class of funtions, or be
considerably more complicated when they must robustly deal with cases of already exe
recursive functions and exceptional returns.

Particular care (and experimentation) is required writing metrics for these cases.

7 CONCLUSION

This paper was written to help you become more proficient in programming with Paradyn’s
ric Description Language (MDL). Specifically, it was designed to save you time and frustra
in using this new and unusual language. We discussed the two basic MDL metric value

counters and timers. We covered tips on how to properly update them in instrumentation. W
covered how to keep metric values from decreasing.

We have tried to keep this document brief and readable so that you may be encouraged
it in its entirety. For more MDL examples, the best source is theparadyn.rc configuration file
provided to you. However, some metrics in theparadyn.rc were implemented using in-depth
knowledge about the functions and/or computing platforms involved, so some of the prog
ming tips developed in this paper were skipped without affecting the correctness of those m
Please keep in mind that such discrepancies do not invalidate the advice given in this pape
you may not have (or want to acquire) such detailed information about your computing plat

To simplify the presentation, we provided only MDL code fragments, and not complete m
implementations. In addition, thoughconstraintsare key to instrumentation, none of the MD
examples given wereconstrained . All examples were written as if for whole-program metric
Again, this was done to simplify the text. The metrics you write, however, will usually need t
constrained .

Thank you for your interest in Paradyn. We hope this paper was useful to you and that
future reading continues to provide you new insights. If you have additional questions, p
send email toparadyn@cs.wisc.edu .

■

